
Als ich vor einiger Zeit aus den USA zurückkam, war ich mit einem sehr ernsten Problem eines unserer Kunden konfrontiert. Er sollte eine Anwendung für unser Auslandsbüro erstellen. Die Anwendung sollte die folgenden Operationen ausführen können:

· Speichern von Daten in Unicode

· Anzeigen von Daten in Unicode

Speichern von Daten in Unicode

Dies war der einfache Teil. Unicode-Zeichen werden intern durch einen Zwei-Byte-Code repräsentiert. Daher können Sie diese Daten einfach in VFPs Feldern vom Typ Zeichen (binär) und Memo (binär) abspeichern. Bedenken Sie dabei, dass Sie für jedes Zeichen zwei Bytes benötigen, so dass Sie die Felder doppelt so groß machen müssen.

Anzeigen von Daten in Unicode

Obwohl Visual FoxPro DBCS unterstützt, ist die Unterstützung für Unicode stark limitiert:

· Es gibt keine in VFP eingebauten Kontrollelemente, die Unicode-Zeichen anzeigen können.

· Sie können keinen String im Format Unicode speichern.

· Wenn Sie den Wert eines Strings einer Variablen zuordnen, wird er automatisch in das Format ANSI konvertiert.

Sie haben also keine Chance, mit den mitgelieferten Möglichkeiten von VFP Unicode-Zeichen zu unterstützen.

Die nativen Textboxen von VFP können lediglich ANSI-Zeichen anzeigen, keine Unicode-Strings. Um dieses Problem zu umgehen, können Sie die MS Forms 2.0-Kontrollelemente einsetzen, die mit dem Office-Paket ausgeliefert werden. Es ist daher erforderlich, über eine Office-Installation auf dem System zu verfügen.

Damit haben wir Kontrollelemente gefunden, die Unicode-Zeichen anzeigen können. Jetzt bleibt uns nur noch das Problem, wie wir die Unicode-Zeichen aus dem Binärformat an die MS Forms Textbox übergeben. VFP verfügt über keine Funktionalität, um Unicode-Zeichen im Speicher zu verwalten. Auch wenn wir am Anfang über Unicode-Zeichen verfügen, werden diese durch eine einfache Zuweisung zu einer Variablen in ANSI-Strings konvertiert. Die neue Funktion COMPROP() löst das Problem nicht. Damit ist ein direkter Einsatz von MS Forms 2.0 nicht möglich.

Auf der anderen Seite ist Visual Basic sehr gut in der Lage, Unicode-Zeichen zu verarbeiten und stellt einen hervorragenden ActiveX-Container dar. Zusätzlich können Sie Kontrollelemente in VB erstellen und anschließend in anderen Anwendungen einsetzen. VB besitzt noch eine andere Fähigkeit, mit der dieses Problem gelöst werden kann. Es kann auf einfache Weise Byte-Arrays in Unicode-Zeichen konvertieren. Damit haben wir einen Weg gefunden, dieses Problem zu lösen.

Die unicodefähigen Kontrollelemente von MS Forms 2.0 werden in Wrapper-Kontrollelemente von VB eingeschlossen, die ihre erforderlichen Methoden an das Formular von VFP exportieren. In VFP werden die Unicode-Strings als Long-Arrays repräsentiert, zwei Werte vom Typ long für ein Unicode-Zeichen. Diese Arrays können sehr einfach in Feldern vom Typ Memo (binär) oder Zeichen (binär) dauerhaft gespeichert werden.

Lösungsweg

Nach der Erklärung der Architektur erkläre ich detailliert die Lösung:

Die ActiveX-Datei UnicodeControls3.ocx enthält die zwei Kontrollelemente utlUnicodeButton3 und utlUnicodeTextbox3.

A) Das Kontrollelement utlUnicodeButton

Als erstes sehen wir uns die Implementierung von utlUnicodeButton3 an.

Dieses Element kann eingesetzt werden, um eine Art virtuelle Tastatur zu erstellen. Sie können die Eigenschaft Caption dieses Kontrollelements mit der Methode SetButtonCaption füllen.

' set button caption to 

' needed unicode char

Public Function SetButtonCaption( _

    ByVal lngUnicodeLoByte As Long, _

    ByVal lngUnicodeHiByte As Long)

    As Long

    Dim strCaption As String

    On Error GoTo ErrorHandler

    ' set caption of command button

    strCaption = ""

    If GetUnicodeString( _

        lngUnicodeLoByte, 

        lngUnicodeHiByte, 

        strCaption) = 

        ERROR_SUCCESS Then

      cmdUnicodeDisplay.Caption = 

        strCaption

    End If

    ' store byte codes in properties

    Me.UnicodeHiByte=lngUnicodeHiByte

    Me.UnicodeLoByte=lngUnicodeLoByte

    SetButtonCaption = ERROR_SUCCESS

    Exit Function

ErrorHandler:

    WriteError "SetButtonCaption", 

        Err.Number, Err.Description

    SetButtonCaption = Err.Number

End Function

Diese Methode erhält zwei Long-Werte und ruft die Funktion GetUnicodeString auf, um aus diesen Werten Unicode zu erzeugen. Die Zwei-Byte-Werte werden in den zwei Eigenschaften UnicodeHiByte und UnicodeLoByte gespeichert, die vom Ereignis ButtonClick, das ich später noch behandeln werde, aus genutzt werden können.

Um arbeiten zu können, muss das Kontrollelement ein Ereignis mit Namen ButtonClick exportieren, das ich von meinem mit VFP erstellten Formular aus nutzen kann.

' define event

Public Event ButtonClick( _

ByRef lngUnicodeLoByte As Long, _

ByRef lngUnicodeHiByte As Long)

' Click event

Private Sub cmdUnicodeDisplay_Click()

    On Error GoTo ErrorHandler

    RaiseEvent ButtonClick(Me.UnicodeLoByte, Me.UnicodeHiByte)

    Exit Sub

ErrorHandler:

    WriteError "cmdUnicodeDisplay_Click()", Err.Number, Err.Description

End Sub

Zunächst definieren wir das neue Ereignis ButtonClick, und im Ereignis Click der Schaltfläche wird ButtonClick mit der Methode RaiseEvent ausgelöst.

Die nächste wichtige Aufgabe ist die Definition der Eigenschaften UnicodeLoByte und UnicodeHiByte. Dies erledigen wir über so genannte Eigenschaftenmethoden, die den Access/Assign-Methoden entsprechen, die VFP seit der Version 6.0 kennt.

' define member variables

Private m_lngUnicodeLoByte As Long

Private m_lngUnicodeHiByte As Long

' property procedures for 'UnicodeLoByte'

Public Property Get UnicodeLoByte() As Long

    UnicodeLoByte = m_lngUnicodeLoByte

End Property

Public Property Let UnicodeLoByte(ByVal vNewValue As Long)

    m_lngUnicodeLoByte = vNewValue

End Property

' property procedures for 'UnicodeHiByte'

Public Property Get UnicodeHiByte() As Variant

    UnicodeHiByte = m_lngUnicodeHiByte

End Property

Public Property Let UnicodeHiByte(ByVal vNewValue As Variant)

    m_lngUnicodeHiByte = vNewValue

End Property

Leider verfügen Sie in der Entwurfsansicht von VFP über keinerlei Zugriff auf die Eigenschaften, so dass Sie sie nur zur Laufzeit mit den unten beschriebenen Funktionen und Eigenschaften einstellen können. 

Die Methode SetButtonPos empfängt die für die Position und Größe erforderlichen Parameter und ordnet diese über die Funktion SetControlPosition den Eigenschaften der Schaltfläche zu

' set button position inside the container

Public Function SetButtonPos( _

lngTop As Long, _

lngLeft As Long, _

lngHeight As Long, _

lngWidth As Long) As Long

    Dim lngRetVal As Long

    On Error GoTo ErrorHandler

lngRetVal = SetControlPosition( _

cmdUnicodeDisplay, lngTop, lngLeft, lngHeight, lngWidth)

    SetButtonPos = ERROR_SUCCESS

ErrorHandler:

    WriteError "SetButtonPos", Err.Number, Err.Description

    SetButtonPos = Err.Number

End Function

Zusätzlich können Sie jede Eigenschaft über ihre korrespondierende Eigenschaft einstellen. Die Eigenschaften ButtonTop, ButtonLeft, ButtonHeight und ButtonWidth exportieren die Eigenschaften des inneren Schaltflächenobjekts für den Lese- und Schreibzugriff von außen.

' property procedures for 'ButtonTop'

Public Property Get ButtonTop() As Variant

    ButtonTop = cmdUnicodeDisplay.Top

End Property

Public Property Let ButtonTop(ByVal vNewValue As Variant)

    cmdUnicodeDisplay.Top = vNewValue

End Property

' property procedures for 'ButtonLeft'

Public Property Get ButtonLeft() As Variant

    ButtonLeft = cmdUnicodeDisplay.Left

End Property

Public Property Let ButtonLeft(ByVal vNewValue As Variant)

    cmdUnicodeDisplay.Left = vNewValue

End Property

' property procedures for 'ButtonHeight'

Public Property Get ButtonHeight() As Variant

    ButtonHeight = cmdUnicodeDisplay.Height

End Property

Public Property Let ButtonHeight(ByVal vNewValue As Variant)

    cmdUnicodeDisplay.Height = vNewValue

End Property

' property procedures for 'ButtonWidth'

Public Property Get ButtonWidth() As Variant

    ButtonWidth = cmdUnicodeDisplay.Width

End Property

Public Property Let ButtonWidth(ByVal vNewValue As Variant)

    cmdUnicodeDisplay.Width = vNewValue

End Property

Wenn sich die Hintergrundfarbe meines Formulars von der im Kontrollelement gewählten Hintergrundfarbe unterscheidet, muss ich die Möglichkeit, haben, zur Laufzeit eine Auswahl zu treffen. Daher habe ich die Eigenschaft ButtonBackColor erstellt, die einen Lese- und Schreibzugriff auf die Eigenschaft BackColor des Kontrollelements eröffnet.

' property procedures for 'ButtonBackColor'

Public Property Get ButtonBackColor() As Variant

    ButtonBackColor = BackColor

End Property

Public Property Let ButtonBackColor(ByVal vNewValue As Variant)

    BackColor = vNewValue

End Propertyv

B) Das Kontrollelement utlUnicodeTextbox

utlUnicodeTextbox ist das zweite Kontrollelement. Es wird eingesetzt, um die Unicode-Strings anzuzeigen und zu ändern.

Die wichtigste Methode ist InsertUnicodeChar. Diese Funktion empfängt zwei Long-Werte, konvertiert diese in Unicode-Zeichen und fügt sie an der Position des Cursors in die Textbox ein.

Public Function InsertUnicodeChar( _

    ByVal lngUnicodeLoByte As Long, _

    ByVal lngUnicodeHiByte As Long) 

       As Long

    Dim strUnicode As String

    Dim strUnicodeLeft As String

    Dim strUnicodeRight As String

    Dim lngRetVal As Long

    Dim intCurPos As Integer

    strUnicode = ""

    ' convert two byte values 

    ' into a unicode char

    lngRetVal = GetUnicodeString( _

        lngUnicodeLoByte, 

        lngUnicodeHiByte, strUnicode)

    If lngRetVal <> ERROR_SUCCESS Then

        InsertUnicodeChar = lngRetVal

        Exit Function

    End If

    ' notice current cursor position

    intCurPos=txtUnicodeTextbox.SelStart

    ' insert the new char in 

    ' the cursor position

    With txtUnicodeTextbox

        strUnicodeLeft = Left(.Text, 

                           intCurPos)

        strUnicodeRight = Right(.Text,

              Len(.Text) - intCurPos)

        .Text = strUnicodeLeft & 

          strUnicode & strUnicodeRight

        .SelStart = intCurPos + 1

    End With

    InsertUnicodeChar = ERROR_SUCCESS

    Exit Function

ErrorHandler:

    WriteError "InsertUnicodeChar",

        Err.Number, Err.Description

    InsertUnicodeChar = Err.Number

End Function

Für diese Aufgabe wird die Funktion GetUnicodeString eingesetzt. Mit der Eigenschaft SelStart ermittle ich die richtige Position für das Einfügen der neuen Unicode-Zeichen.

Die nächste wichtige Methode ist GetByteArray. Sie empfängt ein Variant-Array und füllt es mit dem Inhalt der Textbox. Auf diese Weise empfängt das VFP-Formular ein Array mit Variant-Werten, die den Unicode-String in der Textbox repräsentieren.

Public Function GetByteArray(ByRef arrLongVar() As Variant) As Long

    Dim lngRetVal As Long

    Dim arrLong() As Long

    Dim i As Integer

    On Error GoTo ErrorHandler

    ReDim arrLong(1 To 1)

    ' transfer array from variant 

    ' to long

    For i = 1 To UBound(arrLongVar)

        arrLong(i)=CLng(arrLongVar(i))

    Next

    ' check for empty textbox

    If Len(txtUnicodeTextbox.Text) > 0

      Then

        GetByteArray = _ 

              UnicodeToByteArray(

              txtUnicodeTextbox.Text, 

              arrLong())

    Else

        GetByteArray 

             = ERROR_EMPTY_TEXTBOX

    End If

    ' transfer array values from 

    ' long to unicode

    ReDim arrLongVar(1 To UBound(

                          arrLong))

    For i = 1 To UBound(arrLongVar)

        arrLongVar(i) = arrLong(i)

    Next

    Exit Function

ErrorHandler:

    GetByteArray = Err.Number

End Function

Die Methode setzt die Funktion UnicodeToByteArray ein, die einen Unicode-String sowie ein Array empfängt und das Array mit den Long-Werten füllt, die den Unicode-String repräsentieren.

Für die Steuerung der Position und der Größe der Textbox finden Sie hier die gleichen Funktionen wie in utlUnicodeCommand. Auch die Eigenschaft TextBoxBackColor für den Lese- und Schreibzugriff auf die Eigenschaft BackColor des Kontrollelements habe ich eingefügt.

' set textbox position inside 

' the container

Public Function SetTextBoxPos(lngTop As Long, lngLeft As Long, lngHeight As Long, lngWidth As Long) As Long

    Dim lngRetVal As Long

    On Error GoTo ErrorHandler

    lngRetVal = SetControlPosition( _

        txtUnicodeTextbox, lngTop,

        lngLeft, lngHeight, lngWidth)

    SetTextBoxPos = ERROR_SUCCESS

ErrorHandler:

    WriteError "SetButtonPos",

        Err.Number, Err.Description

    SetTextBoxPos = Err.Number

End Function

' property procedures for 'TextBoxTop'

Public Property Get TextBoxTop() As Variant

    TextBoxTop = txtUnicodeTextbox.Top

End Property

Public Property Let TextBoxTop(ByVal vNewValue As Variant)

    txtUnicodeTextbox.Top = vNewValue

End Property

' property procedures for 'TextBoxLeft'

Public Property Get TextBoxLeft() As Variant

    TextBoxLeft = txtUnicodeTextbox.Left

End Property

Public Property Let TextBoxLeft(ByVal vNewValue As Variant)

    txtUnicodeTextbox.Left = vNewValue

End Property

' property procedures for 'TextBoxHeight'

Public Property Get TextBoxHeight() As Variant

    TextBoxHeight = txtUnicodeTextbox.Height

End Property

Public Property Let TextBoxHeight(ByVal vNewValue As Variant)

    txtUnicodeTextbox.Height = vNewValue

End Property

' property procedures for 'TextBoxWidth'

Public Property Get TextBoxWidth() As Variant

    TextBoxWidth = txtUnicodeTextbox.Width

End Property

Public Property Let TextBoxWidth(ByVal vNewValue As Variant)

    txtUnicodeTextbox.Width = vNewValue

End Property

' property procedures for 'TextBoxBackColor'

Public Property Get TextBoxBackColor() As Variant

    TextBoxBackColor = BackColor

End Property

Public Property Let TextBoxBackColor(ByVal vNewValue As Variant)

    BackColor = vNewValue

End Property

C) Das Modul basGlobal.bas

Beide Klassenmodule nutzen das gemeinsame Modul basGlobal.bas, das einige grundlegende Funktionalitäten für die Fehlerbehandlung sowie die Manipulation der Unicode-Strings enthält.

Dazu gehören auch einige Fehlerkonstanten, mit denen Sie feststellen können, in welcher Methode ein möglicher Fehler aufgetreten ist.

Public Const ERROR_SUCCESS As Long = 0

Public Const ERROR_UNKNOWN As Long =-1

Public Const ERROR_FILL_UNICODE_ARRAY As Long = -4

Public Const ERROR_INSERT_UNICODE_CHAR As Long = -5

Public Const ERROR_GET_BYTE_ARRAY As Long = -6

Public Const ERROR_EMPTY_TEXTBOX As Long = -7

In der Funktion GetUnicodeString werden die beiden Long-Werte den beiden Elementen eines Array zugeordnet. Dieses ByteArray kann einfach einer String-Variablen zugeordnet werden, die in VB schon als Design in Unicode vorhanden ist.

Public Function GetUnicodeString( _

    ByVal lngUnicodeLoByte As Long, _

    ByVal lngUnicodeHiByte As Long, _

    ByRef strUnicode) As Long

    On Error GoTo ErrorHandler

    Dim arrByteUnicode(0 To 1) As Byte

    arrByteUnicode(0) = 

                  lngUnicodeLoByte

    arrByteUnicode(1) = 

                  lngUnicodeHiByte

    strUnicode = arrByteUnicode

    GetUnicodeString = ERROR_SUCCESS

    Exit Function

ErrorHandler:

    WriteError "GetUnicodeString", 

        Err.Number, _

        Err.Description & "--->" & 

        strUnicode

    GetUnicodeString = Err.Number

End Function

Die Funktion UnicodeToByteArray konvertiert einen hereinkommenden Unicode-String in ein Long-Array, das an VFP zurückgeliefert werden kann. Wie schon oben erleichtert das Zuweisen eines Unicode-String an eine Byte-Array die Aufgabe.

Public Function UnicodeToByteArray( _

    strUnicode As String, _

    arrLong() As Long) As Long

    Dim arrByte() As Byte

    Dim i As Integer

    On Error GoTo ErrorHandler

    ' convert unicode to byte array

    arrByte = strUnicode

    ' dimension arrLong

    ReDim arrLong(1 To UBound(

                       arrByte) + 1)

    ' transfer the byte values 

    ' to long values

    For i = 1 To UBound(arrLong)

        arrLong(i) = arrByte(i - 1)

    Next

    ' return value

    UnicodeToByteArray = ERROR_SUCCESS

    Exit Function

ErrorHandler:

    WriteError "UnicodeToByteArray", 

                       Err.Number, _

          Err.Description & "--->" &

                          strUnicode

    UnicodeToByteArray = Err.Number

End Function

Diese Funktion wird nur eingesetzt, um Ereignisse mitzuschreiben.

Public Sub WriteEvent(strErrorMsg As String)

    On Error Resume Next

    App.LogEvent strErrorMsg,

                 vbLogEventTypeError

End Sub

Normalerweise wird diese Funktion eingesetzt, um Ausnahmen oder andere Probleme festzuhalten. Dafür wird die Funktion WriteEvent eingesetzt.

Public Sub WriteError( _

    strProcName As String, lngErrorNo As Long, strErrDesc As String)

    On Error Resume Next

    WriteEvent "Error in " & 

        strProcName & " - " & _

        Trim(Str(lngErrorNo)) & 

        " - " & strErrDesc

End Sub

Diese Funktion wird aufgerufen, um die Position und Größe der Textbox und Schaltfläche einzustellen.

' set control position inside 

' the container

Public Function SetControlPosition(ctrl As Control, lngTop As Long, _

    lngLeft As Long, lngHeight As Long, lngWidth As Long) As Long

    On Error GoTo ErrorHandler

    With ctrl

        .Top = lngTop

        .Left = lngLeft

        .Height = lngHeight

        .Width = lngWidth

    End With

    SetControlPosition = ERROR_SUCCESS

ErrorHandler:

    WriteError "SetControlPosition",

                     Err.Number, _

               ctrl.Name & " - " & 

                   Err.Description

    SetControlPosition = Err.Number

End Function

D) Die VFP-Implementierung

Das waren die Vorbereitungen unter VB. Sehen wir uns jetzt an, wie diese Kontrollelemente auf einem VFP-Formular implementiert werden und wie wir diese anschließend zum Arbeiten bringen.

Öffnen Sie in VFP 7.0 das Projekt „Unicode“. Es sollte auch in früheren Versionen von VFP arbeiten, das habe ich aber nicht getestet. In dem Projekt finden Sie das Formular frmDisplayUnicode3 sowie die Textbox txtUnicode vom Typ utlUnicodeTextbox3.

Die erste Aufgabe, um dies zum Arbeiten zu bringen, ist das Einstellen der Caption der Schaltfläche auf Unicode. Dies geschieht in cmdSetButtonCaption der Schaltfläche.

Sehen sie sich das Ereignis Click an.

LOCAL lngRetVal

lngRetVal = ThisForm.cmdUnicode1.;

  OBJECT.SetButtonCaption(0x39, 0x4E)

lngRetVal = ThisForm.cmdUnicode2.;

       OBJECT.SetButtonCaption(89,2)

Um die erforderlichen Unicode-Zeichen der Schaltfläche anzuzeigen, ist es lediglich erforderlich, die Methode SetButtonCaption auszuführen.

Die nächste Schaltfläche cmdFillTextbox füllt die Textbox.

LOCAL lnRetVal

lnRetVal = ThisForm.txtUnicode.

  Object.InsertUnicodeChar(0x39, 0x4E)

ThisForm.txtUnicode.SetFocus()

lnRetVal = ThisForm.txtUnicode.

   Object.InsertUnicodeChar(0x44, 0x0)

ThisForm.txtUnicode.SetFocus()

Die Schaltfläche nutzt die Methode InsertUnicodeChar() mit zwei numerischen Werten. Sie können diese in einem Array speichern und die Methode InsertUnicodeChar in einer Schleife ausführen.

Auf diese Weise ist es einfach, die Textbox mit Werten zu füllen, die in der Umgebung von VFP gespeichert sind.

Die letzte Schaltfläche cmdGetTextboxValue schreibt den Wert in ein Array.

LOCAL ARRAY laUnicode[1]

LOCAL lngRetVal AS Long

LOCAL lcUnicodeArray as String

LOCAL lnByte as Integer

laUnicode[1] = 0

COMARRAY(ThisForm.txtUnicode.Object,;

                                  11)

lngRetVal = ThisForm.txtUnicode.

      Object.GetByteArray(@laUnicode)

lcUnicodeArray = ""

FOR EACH lnByte IN laUnicode

    lcUnicodeArray = lcUnicodeArray ;

         + ALLTRIM(STR(lnByte)) + " "

NEXT

MESSAGEBOX(lcUnicodeArray)

Die Schaltfläche nutzt im Grunde die Methode GetByteArray() um ein Array zu erhalten, das mit der Unicode-Repräsentation des Inhalts der Textbox gefüllt ist. 

Daraus ergibt sich ein Problem. Wenn ich auf die Unicode-Schaltflächen klicke, will ich die angezeigten Unicode-Zeichen an der aktuellen Position des Cursors in der Textbox anzeigen.

Jetzt benötige ich das Ereignis ButtonClick, das ich der Unicode-Schaltfläche hinzugefügt habe. Wenn Sie sich die Methode von cmdUnicode1 ansehen, werden Sie das Ereignis ButtonClick finden, das ich in utlUnicodeButton3 hinzugefügt habe. Die folgende Ereignisprozedur fügt die angezeigten Unicode-Zeichen an der korrekten Position ein.

*** ActiveX-Steuerelementereignis ***

LPARAMETERS lngunicodelobyte, lngunicodehibyte

LOCAL lnRetVal

lnRetVal = ThisForm.txtUnicode.OBJECT.;

             InsertUnicodeChar(;

             lngunicodelobyte, ;

             lngunicodehibyte)

ThisForm.txtUnicode.SetFocus()

Dies ist die Funktionalität, die erforderlich ist, VFP 7.0 einige Funktionalitäten von Unicode hinzuzufügen.

Aus Gründen des Designs finden Sie einigen Code im Ereignis Init des Formulars:

WITH ThisForm


.txtUnicode.Object. ;



SetTextBoxPos(0, 0, 30, 200)


.cmdUnicode1.Object. ;



SetButtonPos(0, 0, 30, 30)


.cmdUnicode2.Object. ;



SetButtonPos(0, 0, 30, 30)


.cmdUnicode1.Object.;



ButtonBackColor = .BackColor


.cmdUnicode2.Object.;



ButtonBackColor = .BackColor


.txtUnicode.Object.;


   TextBoxBackColor =




 .BackColor

ENDWITH

Mit den Methoden SetTextPos und SetButtonPos werden die Position und Größe des Kontrollelements eingestellt.

Die Eigenschaften ButtonBackColor und TextBackColor werden benötigt, um die Hintergrundfarbe des Kontrollelements auf die Hintergrundfarbe des Formulars einzustellen.

Erweiterungen

Dies ist ein sehr einfaches Beispiel, wie Sie diese Funktionalität nutzen können. Sie können von diesen Kontrollelementen innerhalb von VFP weitere Instanzen ableiten, um sie besser bedienbar zu machen, aber sie enthalten bereits die grundlegenden Funktionalitäten, um dieses Problem zu lösen.

